PrSAT: Some Examples

First, load the **PrSAT** package:

```
<< PrSAT`
```

■ Example #1: *X*, *Y*, *Z* *pairwise* independent, but *not* independent

```
MODEL1 =
 PrSAT[
  {
   Pr[X ∧ Y] == Pr[X] Pr[Y],
   Pr[X ∧ Z] == Pr[X] Pr[Z],
   Pr[Z ∧ Y] == Pr[Z] Pr[Y],
   Pr[X ∧ Y ∧ Z] ≠ Pr[X] Pr[Y] Pr[Z]
  },
  Probabilities → Regular
 ]
```

$$\Big\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\}, Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\},$$

$$\Big\{a_1 \to \frac{295043 - 117\sqrt{4676097}}{168831}, a_2 \to \frac{-84085 + 39\sqrt{4676097}}{168831}, a_3 \to \frac{-84085 + 39\sqrt{4676097}}{168831}, a_4 \to \frac{-84085 + 39\sqrt{4676097}}{168831}, a_5 \to \frac{42}{169}, a_6 \to \frac{42}{169}, a_7 \to \frac{42}{169}, a_8 \to \frac{1}{999}\Big\}\Big\}$$

```
TruthTable[MODEL1]
```

X	Y	Z	var	Pr
T	T	T	a_8	$\frac{1}{999}$
T	T	F	a_5	$\frac{42}{169}$
T	F	T	a_6	$\frac{42}{169}$
T	F	F	a_2	$\frac{-84085+39\sqrt{4676097}}{168831}$
F	T	T	a_7	$\frac{42}{169}$
F	T	F	a_3	$\frac{-84085+39\sqrt{4676097}}{168831}$
F	F	T	a_4	$\frac{-84085+39\sqrt{4676097}}{168831}$
F	F	F	a_1	$\frac{295043-117\sqrt{4676097}}{168831}$

```
EvaluateProbability[
 {
  Pr[X ∧ Y] == Pr[X] Pr[Y],
  Pr[X ∧ Z] == Pr[X] Pr[Z],
  Pr[Z ∧ Y] == Pr[Z] Pr[Y],
  Pr[X ∧ Y ∧ Z] ≠ Pr[X] Pr[Y] Pr[Z]
 }, MODEL1]
```

```
{True, True, True, True}
```

■ Example #2: *s* violates (†)

```
s[h_, e_] := Pr[h | e] - Pr[h | ¬ e];
```

```
MODEL2 =
 PrSAT[
  {
   Pr[ℍ | 𝔼1] > Pr[ℍ | 𝔼2],
   s[ℍ, 𝔼1] < s[ℍ, 𝔼2]
  },
  Probabilities → Regular]
```

$\{\{\mathbb{E}1 \to \{a_2, a_5, a_6, a_8\},$
$\mathbb{E}2 \to \{a_3, a_5, a_7, a_8\}, \mathbb{H} \to \{a_4, a_6, a_7, a_8\},$
$\Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\},$
$\{a_1 \to \frac{510\,368\,209}{3\,344\,528\,124}, a_2 \to \frac{2}{31}, a_3 \to \frac{4}{21}, a_4 \to \frac{1}{999},$
$a_5 \to \frac{1}{49}, a_6 \to \frac{1}{532}, a_7 \to \frac{10}{37}, a_8 \to \frac{26}{87}\}\}$

```
TruthTable[MODEL2]
```

$\mathbb{E}1$	$\mathbb{E}2$	$\mathbb{H}$	var	Pr
T	T	T	a_8	$\frac{26}{87}$
T	T	F	a_5	$\frac{1}{49}$
T	F	T	a_6	$\frac{1}{532}$
T	F	F	a_2	$\frac{2}{31}$
F	T	T	a_7	$\frac{10}{37}$
F	T	F	a_3	$\frac{4}{21}$
F	F	T	a_4	$\frac{1}{999}$
F	F	F	a_1	$\frac{510\,368\,209}{3\,344\,528\,124}$

```
EvaluateProbability[
 {
  Pr[ℍ | 𝔼1] > Pr[ℍ | 𝔼2],
  s[ℍ, 𝔼1] < s[ℍ, 𝔼2]
 }, MODEL2]
```

{True, True}

Note: *d satisfies* (†)

```
d[h_, e_] := Pr[h | e] - Pr[h];

PrSAT[
 {
  Pr[ℍ | 𝔼1] ≥ Pr[ℍ | 𝔼2],
  d[ℍ, 𝔼1] < d[ℍ, 𝔼2]
 }
]
```

— *PrSAT::srchfail :*
Search phase failed; attempting
FindInstance

```
{}
```

■ Example #3: A Counterexample Requiring 4-Events (Douven)

Question: Given a finite set *S* of propositions such that (1) each proposition *H* in *S* is correlated with the conjunction of the members of any non-empty of subset of *S* that does not contain *H*, is (2) the conjunction of the members of any non-empty subset *S*' of *S* correlated with the conjunction of any non-empty subset *S*'' of *S* not overlapping with *S*'?

Answer: **No**. Note: the answer is YES for 3-event spaces. So, we assume a four-event probabilitiy space, over the propositions {*X*, *Y*, *Z*, *W*}. To try to find a 4-element countermodel, we first represent condition (1):

```
C1 :=
  {
   Pr[X | Y] > Pr[X],
   Pr[X | Z] > Pr[X],
   Pr[X | W] > Pr[X],
   Pr[Y | Z] > Pr[Y],
   Pr[Y | W] > Pr[Y],
   Pr[Z | W] > Pr[Z],
   Pr[X | Y ∧ Z] > Pr[X],
   Pr[X | Y ∧ W] > Pr[X],
   Pr[X | Z ∧ W] > Pr[X],
   Pr[Y | X ∧ Z] > Pr[Y],
   Pr[Y | X ∧ W] > Pr[Y],
   Pr[Y | Z ∧ W] > Pr[Y],
   Pr[Z | X ∧ Y] > Pr[Z],
   Pr[Z | X ∧ W] > Pr[Z],
   Pr[Z | Y ∧ W] > Pr[Z],
   Pr[X | Y ∧ Z ∧ W] > Pr[X],
   Pr[Y | X ∧ Z ∧ W] > Pr[Y],
   Pr[Z | X ∧ Y ∧ W] > Pr[Z]
  };
```

Now, the key will be trying to find models in which **`C1`** (*i.e.*, all the inequalities in **`C1`**) holds, and in which at least one of the following equalities holds (these would, of course, be models in which (2) *fails*):

```
C2 :=
 {
  Pr[X ∧ Y | Z ∧ W] == Pr[X ∧ Y],
  Pr[X ∧ Z | Y ∧ W] == Pr[X ∧ Z],
  Pr[Y ∧ Z | X ∧ W] == Pr[Y ∧ Z]
 }
```

In fact, we'll find a model in which **all** the equations in **`C2`** hold, using our **`PrSAT`** function, as follows:

```
MODEL3 =
 PrSAT[

  Union[C1, C2],

  Probabilities → Regular
 ]
```

$$\{\{W \to \{a_2, a_6, a_7, a_8, a_{12}, a_{13}, a_{14}, a_{16}\},$$
$$X \to \{a_3, a_6, a_9, a_{10}, a_{12}, a_{13}, a_{15}, a_{16}\},$$
$$Y \to \{a_4, a_7, a_9, a_{11}, a_{12}, a_{14}, a_{15}, a_{16}\},$$
$$Z \to \{a_5, a_8, a_{10}, a_{11}, a_{13}, a_{14}, a_{15}, a_{16}\},$$
$$\Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8, a_9, a_{10}, a_{11}, a_{12}, a_{13}, a_{14}, a_{15}, a_{16}\}\},$$
$$\{a_1 \to \frac{53\,930\,355\,133\,083\,251\,854}{389\,885\,438\,824\,209\,942\,075}, a_2 \to \frac{2}{45},$$
$$a_3 \to \frac{2}{45}, a_4 \to \frac{1}{25}, a_5 \to \frac{2}{45}, a_6 \to \frac{91\,386\,781}{2\,219\,960\,050},$$
$$a_7 \to \frac{372\,759}{8\,066\,450}, a_8 \to \frac{13\,109\,319}{290\,102\,450},$$
$$a_9 \to \frac{1}{18}, a_{10} \to \frac{3}{55}, a_{11} \to \frac{2}{33}, a_{12} \to \frac{1}{14},$$
$$a_{13} \to \frac{4}{55}, a_{14} \to \frac{3}{38}, a_{15} \to \frac{5}{61}, a_{16} \to \frac{2}{25}\}\}$$

```
TruthTable[MODEL3]
```

W	X	Y	Z	var	Pr
T	T	T	T	a_{16}	$\frac{2}{25}$
T	T	T	F	a_{12}	$\frac{1}{14}$
T	T	F	T	a_{13}	$\frac{4}{55}$
T	T	F	F	a_{6}	$\frac{91\,386\,781}{2\,219\,960\,050}$
T	F	T	T	a_{14}	$\frac{3}{38}$
T	F	T	F	a_{7}	$\frac{372\,759}{8\,066\,450}$
T	F	F	T	a_{8}	$\frac{13\,109\,319}{290\,102\,450}$
T	F	F	F	a_{2}	$\frac{2}{45}$
F	T	T	T	a_{15}	$\frac{5}{61}$
F	T	T	F	a_{9}	$\frac{1}{18}$
F	T	F	T	a_{10}	$\frac{3}{55}$
F	T	F	F	a_{3}	$\frac{2}{45}$
F	F	T	T	a_{11}	$\frac{2}{33}$
F	F	T	F	a_{4}	$\frac{1}{25}$
F	F	F	T	a_{5}	$\frac{2}{45}$
F	F	F	F	a_{1}	$\frac{53\,930\,355\,133\,083\,251\,854}{389\,885\,438\,824\,209\,942\,075}$

```
EvaluateProbability[
 {
  Union[C1, C2]
 }, MODEL3]
```

```
{{True, True, True, True, True, True, True,
  True, True, True, True, True, True, True,
  True, True, True, True, True, True, True}}
```

■ "Proving" Theorems

Bayes's Theorem is trivial:

```
PrSAT[
 {
  Pr[X | Y] ≠ Pr[Y | X] Pr[X] / (Pr[Y | X] Pr[X] + Pr[Y | ¬ X] Pr[¬ X])
 }
]
```

$$\mathrm{PrSAT}\left[\left\{\Pr[X \mid Y] \neq \frac{\Pr[Y \mid X]\Pr[X]}{\Pr[Y \mid X]\Pr[X] + \Pr[Y \mid \neg X]\Pr[\neg X]}\right\}\right]$$

— *PrSAT::srchfail :*
Search phase failed; attempting FindInstance

{}

Popper-Miller Theorem "Additivity of *d*-Confirmation" Theorem is trivial:

```
PrSAT[
 {
  d[ℍ, 𝔼] ≠ d[ℍ ∨ 𝔼, 𝔼] + d[ℍ ∨ ¬ 𝔼, 𝔼]
 }
]
```

— *PrSAT::srchfail :*
Search phase failed; attempting FindInstance

{}

```
PrSAT[
 {
  d[ℍ ∨ ¬ 𝔼, 𝔼] > 0
 }
]
```

— *PrSAT::srchfail :*
Search phase failed; attempting
FindInstance

{}

```
PrSAT[
 {
  d[ℍ, 𝔼] ≠ d[ℍ ∧ 𝔼, 𝔼] + d[ℍ ∧ ¬ 𝔼, 𝔼]
 }
]
```

— *PrSAT::srchfail :*
Search phase failed; attempting
FindInstance

{}

```
PrSAT[
 {
  d[ℍ ∧ ¬ 𝔼, 𝔼] > 0
 }
]
```

— *PrSAT::srchfail :*
Search phase failed; attempting
FindInstance

{}

Note: Popper-Miller Theorem *fails* for *r* (first pointed out by Redhead)

```
r[h_, e_] := Pr[h | e] / Pr[h];
```

```
MODEL4 = PrSAT[
  {
   r[ℍ, 𝔼] ≠ r[ℍ ∨ 𝔼, 𝔼] + r[ℍ ∨ ¬ 𝔼, 𝔼]
  },
  Probabilities → Regular
 ]
```

$$\left\{\{\mathbb{E} \to \{a_2, a_4\}, \mathbb{H} \to \{a_3, a_4\}, \Omega \to \{a_1, a_2, a_3, a_4\}\}, \left\{a_1 \to \frac{427}{285714}, a_2 \to \frac{1}{999}, a_3 \to \frac{1}{999}, a_4 \to \frac{285}{286}\right\}\right\}$$

```
TruthTable[MODEL4]
```

$\mathbb{E}$	$\mathbb{H}$	var	Pr
T	T	a_4	$\frac{285}{286}$
T	F	a_2	$\frac{1}{999}$
F	T	a_3	$\frac{1}{999}$
F	F	a_1	$\frac{427}{285714}$

Wagner's "Probabilistic Modus Tollens" Theorem is trivial

$$\mathbf{PrSAT}\left[\left\{\frac{1-\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]-\mathbf{Pr}[\neg\mathbb{E}]}{1-\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]}\geq\frac{\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]+\mathbf{Pr}[\neg\mathbb{E}]-1}{\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]},\ \frac{1-\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]-\mathbf{Pr}[\neg\mathbb{E}]}{1-\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]}>\mathbf{Pr}[\neg\mathbb{H}]\right\}\right]$$

— *PrSAT::srchfail :*
Search phase failed; attempting FindInstance

{}

$$\mathbf{PrSAT}\left[\left\{\frac{\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]+\mathbf{Pr}[\neg\mathbb{E}]-1}{\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]}\geq\frac{1-\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]-\mathbf{Pr}[\neg\mathbb{E}]}{1-\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]},\ \frac{\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]+\mathbf{Pr}[\neg\mathbb{E}]-1}{\mathbf{Pr}[\mathbb{E}\mid\mathbb{H}]}>\mathbf{Pr}[\neg\mathbb{H}]\right\}\right]$$

— *PrSAT::srchfail :*
Search phase failed; attempting FindInstance

{}